Affine Connection and Curvature

The Friedmann Metric

Initialization

In[ ]:=
```
Needs["xAct`xCoba`"]
```

```
------------------------------------------------------------

Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}

CopyRight (C) 2003-2018, Jose M. Martin-Garcia, under the General Public License.

Connecting to external linux executable...

Connection established.

------------------------------------------------------------

Package xAct`xTensor`  version 1.1.3, {2018, 2, 28}

CopyRight (C) 2002-2018, Jose M. Martin-Garcia, under the General Public License.

------------------------------------------------------------

Package xAct`xCoba`  version 0.8.4, {2018, 2, 28}

CopyRight (C) 2005-2018, David Yllanes and
  Jose M. Martin-Garcia, under the General Public License.

------------------------------------------------------------

These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.

------------------------------------------------------------
```

In[ ]:=
```
(*$PrePrint=ScreenDollarIndices;
$DefInfoQ=False;
$UndefInfoQ=False;*)
```

In[ ]:=
```
DefManifold[M, 4, {λ, μ, ν, ρ, σ, α, β, γ, δ}];
dimM = DimOfManifold[M];
dimM1 = dimM - 1;
```

```
** DefManifold: Defining manifold M.

** DefVBundle: Defining vbundle TangentM.
```

Here is the definition of the coordinate system, and the metric:

Simply define the list coords = {....} and the matrix gmatrix

In[ ]:=

```
coords = {t[], χ[], θ[], ϕ[]};
(*DefConstantSymbol[mass,PrintAs→"M"];*)
DefScalarFunction[ascale, PrintAs → "a"];
(*Use as e.g. ascale[t[],r[]] for a function of (t,r)*)
gmatrix = DiagonalMatrix[
   {-1, ascale[t[]]^2, ascale[t[]]^2 Sin[χ[]]^2, ascale[t[]]^2 Sin[χ[]]^2 Sin[θ[]]^2}
  ];
DefChart[ch, M, {0, 1, 2, 3}, coords, ChartColor → Blue];
g = CTensor[gmatrix, {-ch, -ch}];
SetCMetric[g, ch, SignatureOfMetric → {3, 1, 0}];
CD = CovDOfMetric[g];
```

```
** DefScalarFunction: Defining scalar function ascale.
** DefChart: Defining chart ch.
** DefTensor: Defining coordinate scalar t[].
** DefTensor: Defining coordinate scalar χ[].
** DefTensor: Defining coordinate scalar θ[].
** DefTensor: Defining coordinate scalar ϕ[].
** DefMapping: Defining mapping ch.
** DefMapping: Defining inverse mapping ich.
** DefTensor: Defining mapping differential tensor dich[-a, ichα].
** DefTensor: Defining mapping differential tensor dch[-α, cha].
** DefBasis: Defining basis ch. Coordinated basis.
** DefCovD: Defining parallel derivative PDch[-α].
** DefTensor: Defining vanishing torsion tensor TorsionPDch[α, -β, -γ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelPDch[α, -β, -γ].
** DefTensor: Defining vanishing Riemann tensor RiemannPDch[-α, -β, -γ, δ].
** DefTensor: Defining vanishing Ricci tensor RicciPDch[-α, -β].
** DefTensor: Defining antisymmetric +1 density etaUpch[α, β, γ, δ].
** DefTensor: Defining antisymmetric -1 density etaDownch[-α, -β, -γ, -δ].
```

In[ ]:=
```
Print[
 "gμν= ", ComponentArray[g[{-μ, -ch}, {-ν, -ch}]] // MatrixForm, "     ,         ",
 "g^μν= ", ComponentArray[g[{ μ,  ch}, { ν,  ch}]] // MatrixForm, "\n",
 "g = ", Determinant[g, ch][], " = ",
 Det[ComponentArray[g[{-μ, -ch}, {-ν, -ch}]]] // Simplify
]
```

$$g_{\mu\nu}= \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & a[t]^2 & 0 & 0 \\ 0 & 0 & a[t]^2\,\mathrm{Sin}[\chi]^2 & 0 \\ 0 & 0 & 0 & a[t]^2\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2 \end{pmatrix} \quad , \qquad g^{\mu\nu}= \begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & \frac{1}{a[t]^2} & 0 & 0 \\ 0 & 0 & \frac{\mathrm{Csc}[\chi]^2}{a[t]^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[\theta]^2\,\mathrm{Csc}[\chi]^2}{a[t]^2} \end{pmatrix}$$

$$g = -a[t]^6\ \mathrm{Sin}[\theta]^2\ \mathrm{Sin}[\chi]^4 = -a[t]^6\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^4$$

Affine Connection

Print nonzero components:

$\Gamma^{\lambda}{}_{\mu\nu} = \frac{1}{2}\, g^{\lambda\sigma}(\partial_\mu\, g_{\sigma\nu} + \partial_\nu\, g_{\sigma\mu} - \partial_\sigma\, g_{\mu\nu})$

The components of the Christoffel symbols are collected by the ComponentArray[expr] function.

In[ ]:=
```
list = ComponentArray[Christoffel[CD, PDch][{α, ch}, {-β, -ch}, {-γ, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k]], 0],
    {Subscript[Superscript["Γ", i - 1], j - 1, k - 1], list[[i, j, k]]}
   ],
   {i, 1, Length[list]}, {j, 1, Length[list]}, {k, 1, j}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$\Gamma^0{}_{1,1}$	$a[t]\, a'[t]$
$\Gamma^0{}_{2,2}$	$a[t]\, \mathrm{Sin}[\chi]^2\, a'[t]$
$\Gamma^0{}_{3,3}$	$a[t]\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\, a'[t]$
$\Gamma^1{}_{1,0}$	$\frac{a'[t]}{a[t]}$
$\Gamma^1{}_{2,2}$	$-\mathrm{Cos}[\chi]\, \mathrm{Sin}[\chi]$
$\Gamma^1{}_{3,3}$	$-\mathrm{Cos}[\chi]\, \mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]$
$\Gamma^2{}_{2,0}$	$\frac{a'[t]}{a[t]}$
$\Gamma^2{}_{2,1}$	$\mathrm{Cot}[\chi]$
$\Gamma^2{}_{3,3}$	$-\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$
$\Gamma^3{}_{3,0}$	$\frac{a'[t]}{a[t]}$
$\Gamma^3{}_{3,1}$	$\mathrm{Cot}[\chi]$
$\Gamma^3{}_{3,2}$	$\mathrm{Cot}[\theta]$

Curvature

Print nonzero components of Riemann: $R^{\mu}{}_{\nu\rho\sigma}$

$R^{\lambda}{}_{\rho\mu\nu} = \partial_{\mu}\, \Gamma^{\lambda}{}_{\nu\rho} - \partial_{\nu}\, \Gamma^{\lambda}{}_{\mu\rho} + \Gamma^{\lambda}{}_{\mu\sigma}\, \Gamma^{\sigma}{}_{\nu\rho} - \Gamma^{\lambda}{}_{\nu\sigma}\, \Gamma^{\sigma}{}_{\mu\rho}$ (Carroll+Hartle’s convention)

xCoba has Wald’s convention, which for a Levi-Civita Connection gives the same reult after some index raising/lowering.

The components of the Riemann tensor are collected by the ComponentArray[expr] function.

In[ ]:=
```
list = ComponentArray[Riemann[CD][{α, ch}, {-β, -ch}, {-γ, -ch}, {-δ, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k, l]], 0], {Subscript[Superscript["R", i - 1], j - 1, k - 1, l - 1]
     , list[[i, j, k, l]]}
   ],
   {i, 1, Length[list]}, {j, 1, Length[list]}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R^{0}{}_{1,1,0}$	$-a[t]\,a''[t]$
$R^{0}{}_{2,2,0}$	$-a[t]\,\mathrm{Sin}[\chi]^2\,a''[t]$
$R^{0}{}_{3,3,0}$	$-a[t]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\,a''[t]$
$R^{1}{}_{0,1,0}$	$-\frac{a''[t]}{a[t]}$
$R^{1}{}_{2,2,1}$	$-\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)$
$R^{1}{}_{3,3,1}$	$-\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)$
$R^{2}{}_{0,2,0}$	$-\frac{a''[t]}{a[t]}$
$R^{2}{}_{1,2,1}$	$1+a'[t]^2$
$R^{2}{}_{3,3,2}$	$-\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)$
$R^{3}{}_{0,3,0}$	$-\frac{a''[t]}{a[t]}$
$R^{3}{}_{1,3,1}$	$1+a'[t]^2$
$R^{3}{}_{2,3,2}$	$\mathrm{Sin}[\chi]^2\left(1+a'[t]^2\right)$

The Riemann tensor with all lower indices: $R_{\mu\nu\rho\sigma}$

In[ ]:=

```
list = ComponentArray[Riemann[CD][{-α, -ch}, {-β, -ch}, {-γ, -ch}, {-δ, -ch}]];
list = Table[
   If[
    UnsameQ[list⟦i, j, k, l⟧, 0], {Subscript["R", i - 1, j - 1, k - 1, l - 1]
      , list⟦i, j, k, l⟧}
   ],
   {i, 1, Length[list]}, {j, 1, i - 1}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R_{1,0,1,0}$	$-a[t]\,a''[t]$
$R_{2,0,2,0}$	$-a[t]\,Sin[\chi]^2\,a''[t]$
$R_{2,1,2,1}$	$a[t]^2\,Sin[\chi]^2\,(1 + a'[t]^2)$
$R_{3,0,3,0}$	$-a[t]\,Sin[\theta]^2\,Sin[\chi]^2\,a''[t]$
$R_{3,1,3,1}$	$a[t]^2\,Sin[\theta]^2\,Sin[\chi]^2\,(1 + a'[t]^2)$
$R_{3,2,3,2}$	$a[t]^2\,Sin[\theta]^2\,Sin[\chi]^4\,(1 + a'[t]^2)$

The Riemann tensor with all upper indices: $R^{\mu\nu\rho\sigma}$

In[ ]:=

```
list = ComponentArray[Riemann[CD][{α, ch}, {β, ch}, {γ, ch}, {δ, ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k, l]], 0],
    {Superscript[Superscript[Superscript[Superscript["R", i - 1], j - 1], k - 1], l - 1]
     , list[[i, j, k, l]]}
   ],
   {i, 1, Length[list]}, {j, 1, i - 1}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R^{1^{0^{1^{0}}}}$	$-\frac{a''[t]}{a[t]^3}$
$R^{2^{0^{2^{0}}}}$	$-\frac{Csc[\chi]^2\, a''[t]}{a[t]^3}$
$R^{2^{1^{2^{1}}}}$	$\frac{Csc[\chi]^2\,(1+a'[t]^2)}{a[t]^6}$
$R^{3^{0^{3^{0}}}}$	$-\frac{Csc[\theta]^2\, Csc[\chi]^2\, a''[t]}{a[t]^3}$
$R^{3^{1^{3^{1}}}}$	$\frac{Csc[\theta]^2\, Csc[\chi]^2\,(1+a'[t]^2)}{a[t]^6}$
$R^{3^{2^{3^{2}}}}$	$\frac{Csc[\theta]^2\, Csc[\chi]^4\,(1+a'[t]^2)}{a[t]^6}$

The Ricci tensor: $R_{\mu\nu}$

In[ ]:=

```
list = ComponentArray[Ricci[CD][{-α, -ch}, {-β, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j]], 0], {Subscript["R", i - 1, j - 1]
     , list[[i, j]]}
   ],
   {i, 1, Length[list]}, {j, 1, i}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R_{0,0}$	$-\frac{3\, a''[t]}{a[t]}$
$R_{1,1}$	$2 + 2\, a'[t]^2 + a[t]\, a''[t]$
$R_{2,2}$	$Sin[\chi]^2\,(2 + 2\, a'[t]^2 + a[t]\, a''[t])$
$R_{3,3}$	$Sin[\theta]^2\, Sin[\chi]^2\,(2 + 2\, a'[t]^2 + a[t]\, a''[t])$

Ricci scalar:

In[ ]:=
```
Print["R = ", RicciScalar[CD][]]
```

$$R = \frac{6\left(1 + a'[t]^2 + a[t]\, a''[t]\right)}{a[t]^2}$$

R^2 scalar

In[ ]:=
```
Print["R^2 = ", Kretschmann[CD][]]
```

$$R^2 = \frac{12\left(\left(1 + a'[t]^2\right)^2 + a[t]^2\, a''[t]^2\right)}{a[t]^4}$$

Einstein tensor: $G_{\mu\nu}$

In[ ]:=
```
list = ComponentArray[Einstein[CD][{-α, -ch}, {-β, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j]], 0], {Subscript["G", i - 1, j - 1]
      , list[[i, j]]}
   ],
   {i, 1, Length[list]}, {j, 1, i}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$G_{0,0}$	$\frac{3\left(1+a'[t]^2\right)}{a[t]^2}$
$G_{1,1}$	$-1 - a'[t]^2 - 2\, a[t]\, a''[t]$
$G_{2,2}$	$-\mathrm{Sin}[\chi]^2\left(1 + a'[t]^2 + 2\, a[t]\, a''[t]\right)$
$G_{3,3}$	$-\mathrm{Sin}[\theta]^2\, \mathrm{Sin}[\chi]^2\left(1 + a'[t]^2 + 2\, a[t]\, a''[t]\right)$

Weyl tensor:

In[∘]:=
```
list = ComponentArray[Weyl[CD][{-α, -ch}, {-β, -ch}, {-γ, -ch}, {-δ, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k, l]], 0], {Subscript["C", i-1, j-1, k-1, l-1]
      , list[[i, j, k, l]]}
   ],
   {i, 1, Length[list]}, {j, 1, i-1}, {k, 1, Length[list]}, {l, 1, k-1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[∘]//TableForm=

{}

Geodesic Equations

In[∘]:=
```
DefTensor[u[μ], M];
```

** DefTensor: Defining tensor u[μ].

The geodesic equations are:

$\frac{du^\mu}{d\tau} + \Gamma^\mu{}_{\nu\rho}\, u^\nu\, u^\rho = 0$

geqs is a list with the second term geqs[μ+1]=$\Gamma^\mu{}_{\nu\rho}\, u^\nu\, u^\rho$ for each μ=0,..,d-1.

(careful, in the expressions below, when e.g. $u^{3^2} = (u^3)^2$, the squared terms don't appear nicely)

In[∘]:=
```
geqs = ComponentArray[Christoffel[CD, PDch][{μ, ch}, {-ν, -ch}, {-ρ, -ch}]
        u[{ν, ch}] u[{ρ, ch}]] // ContractBasis // Simplify;
For[i = 1, i ≤ Length[geqs], i++, Print["d/dτ", u[{i - 1, ch}], "+(", geqs[[i]], ")=0"]]
```

$\frac{d}{d\tau} u^0 + (a[t] \left(u^{1^2} + \mathrm{Sin}[\chi]^2 \left(u^{2^2} + \mathrm{Sin}[\theta]^2\, u^{3^2} \right) \right) a'[t]) = 0$

$\frac{d}{d\tau} u^1 + (-\mathrm{Cos}[\chi]\, \mathrm{Sin}[\chi] \left(u^{2^2} + \mathrm{Sin}[\theta]^2\, u^{3^2} \right) + \frac{2\, u^0\, u^1\, a'[t]}{a[t]}) = 0$

$\frac{d}{d\tau} u^2 + (2\, \mathrm{Cot}[\chi]\, u^1\, u^2 - \mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]\, u^{3^2} + \frac{2\, u^0\, u^2\, a'[t]}{a[t]}) = 0$

$\frac{d}{d\tau} u^3 + (\frac{2\, u^3 \left(a[t] \left(\mathrm{Cot}[\chi]\, u^1 + \mathrm{Cot}[\theta]\, u^2 \right) + u^0\, a'[t] \right)}{a[t]}) = 0$

In[ ]:=
```
DefScalarFunction[dt , PrintAs → "ṫ"];
DefScalarFunction[dχ , PrintAs → "χ̇"];
DefScalarFunction[dθ , PrintAs → "θ̇"];
DefScalarFunction[dϕ , PrintAs → "ϕ̇"];
DefScalarFunction[ddt, PrintAs → "ẗ"];
DefScalarFunction[ddχ, PrintAs → "χ̈"];
DefScalarFunction[ddθ, PrintAs → "θ̈"];
DefScalarFunction[ddϕ, PrintAs → "ϕ̈"];
u  = CTensor[{dt[] , dχ[] , dθ[] , dϕ[] }, {ch}];
du = CTensor[{ddt[], ddχ[], ddθ[], ddϕ[]}, {ch}];
{u, du}
```

```
** DefScalarFunction: Defining scalar function dt.
** DefScalarFunction: Defining scalar function dχ.
** DefScalarFunction: Defining scalar function dθ.
** DefScalarFunction: Defining scalar function dϕ.
** DefScalarFunction: Defining scalar function ddt.
** DefScalarFunction: Defining scalar function ddχ.
** DefScalarFunction: Defining scalar function ddθ.
** DefScalarFunction: Defining scalar function ddϕ.
```

Out[ ]=
```
{CTensor[{ṫ[], χ̇[], θ̇[], ϕ̇[]}, {ch}, 0], CTensor[{ẗ[], χ̈[], θ̈[], ϕ̈[]}, {ch}, 0]}
```

In[ ]:=
```
geqs = ComponentArray[Christoffel[CD, PDch][{μ, ch}, {-ν, -ch}, {-ρ, -ch}]
       u[{ν, ch}] u[{ρ, ch}]] // ContractBasis // Simplify;
For[i = 1, i ≤ Length[geqs], i++, Print[du[{i - 1, ch}], "+(", geqs[[i]], ")=0"]]
```

ẗ[]+(a[t] $\left(\dot{\chi}[]^2 + \left(\dot{\theta}[]^2 + \dot{\phi}[]^2\, \mathrm{Sin}[\theta]^2\right) \mathrm{Sin}[\chi]^2\right)$ a′[t])=0

χ̈[]+($-\mathrm{Cos}[\chi] \left(\dot{\theta}[]^2 + \dot{\phi}[]^2\, \mathrm{Sin}[\theta]^2\right) \mathrm{Sin}[\chi] + \dfrac{2\, \dot{t}[]\, \dot{\chi}[]\, a'[t]}{a[t]}$)=0

θ̈[]+($2\, \mathrm{Cot}[\chi]\, \dot{\theta}[]\, \dot{\chi}[] - \mathrm{Cos}[\theta]\, \dot{\phi}[]^2\, \mathrm{Sin}[\theta] + \dfrac{2\, \dot{t}[]\, \dot{\theta}[]\, a'[t]}{a[t]}$)=0

ϕ̈[]+($2\, \dot{\phi}[] \left(\mathrm{Cot}[\theta]\, \dot{\theta}[] + \mathrm{Cot}[\chi]\, \dot{\chi}[] + \dfrac{\dot{t}[]\, a'[t]}{a[t]}\right)$)=0

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year

undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr
Web: http://physics.ntua.gr/konstant